```
SetDirectory[NotebookDirectory[]];
Get["PlanarLinkages.m "];
```

PlanarLinkages — Copyright © 2015 Christoph Koutschan

Demo

Arithmetic of motion polynomials

```
(* Motion polynomials are input using the symbol "eta". *)
t + I + eta * (2 - I)
```

$(i + t) + \eta \cdot (2 - i)$

```
(* When Times is used, eta is always interpreted to be on the left. *)
{I * eta, eta * I}
```

$\{0 + \eta \cdot i, 0 + \eta \cdot i\}$

```
(* To specify the order of factors, use NonCommutativeMultiply . *)
{I ** eta, eta ** I}
```

$\{0 - \eta \cdot i, 0 + \eta \cdot i\}$

```
(* In the same fashion, motion polynomials can be multiplied. *)
(t + 1 - eta) ** (t^2 + 1 + 2 I + eta * I * t)
```

$((1 + 2 i) + (1 + 2 i) t + t^2 + t^3) + \eta \cdot ((-1 - 2 i) + i t - (1 - i) t^2)$

```
(* See how motion polynomials are represented internally. *)
z + eta * w
FullForm [%]
```

$z + \eta \cdot w$

MP[z, w]

```
(* Check the multiplication of motion polynomials (a+eta*b)*(c+eta*d). *)
MP[z1, w1] ** MP[z2, w2]
```

$z_1 z_2 + \eta \cdot (\text{Conjugate}[z_1] w_2 + w_1 z_2)$

```
(* Invert an element in SE2. Note that real numbers in K correspond to 1 in SE2. *)
FullSimplify[(z + eta * w) ** (Conjugate[z] - eta * w)]
```

$\text{Abs}[z]^2 + \eta \cdot 0$

Action of SE_2

```
?ActR2
```

ActR2[p, {x, y}] computes the result of applying the direct isometry
given by p to the point {x, y}, where p is a (possibly constant) motion polynomial .

```
(* Action of SE2. *)
vec = ActR2[MP[x1 + I * x2, y1 + I * y2], {x, y}];
Together[Simplify[vec, Element [{x1, x2, y1, y2}, Reals]]]
```

$$\left\{\frac{x x_1^2 - 2 y x_1 x_2 - x x_2^2 + x_1 y_1 - x_2 y_2}{x_1^2 + x_2^2}, \frac{y x_1^2 + 2 x x_1 x_2 - y x_2^2 + x_2 y_1 + x_1 y_2}{x_1^2 + x_2^2}\right\}$$

```
(* Product in SE2 is bilinear *)
prod = ComplexExpand /@ (MP[x1 + I * x2, y1 + I * y2] ** MP[x1' + I * x2', y1' + I * y2'])
```

$(x_1 x_1' - x_2 x_2' + i (x_2 x_1' + x_1 x_2')) + \eta \cdot (y_1 x_1' - y_2 x_2' + x_1 y_1' + x_2 y_2' + i (y_2 x_1' + y_1 x_2' - x_2 y_1' + x_1 y_2'))$

```
TableForm [Flatten[CoefficientList[List@@prod /. I → i, i]]]
```

$x_1 x_1' - x_2 x_2'$
$x_2 x_1' + x_1 x_2'$
$y_1 x_1' - y_2 x_2' + x_1 y_1' + x_2 y_2'$
$y_2 x_1' + y_1 x_2' - x_2 y_1' + x_1 y_2'$

Fixpoint of a motion

```
? FixPoint
```

FixPoint [p] computes the fixpoint of a simple revolution given by the linear normed motion polynomial p, i.e., p is of the form t – z + η*w where z and w are complex numbers with z being non–real.

```
(* Compute the fixpoint of a motion . *)
FixPoint[t + I - eta * 0]
```

{0, 0}

```
FixPoint[t + eta]
```

Throw::nocatch : Uncaught Throw[FixPoint: Input is not a normed bounded motion polynomial of degree 1.] returned to top level. >>

Hold[Throw[FixPoint: Input is not a normed bounded motion polynomial of degree 1.]]

```
(* Test fixpoint formula . *)
FixPoint[t - α - eta * β]
FullSimplify [(First[%] + I * Last[%]) - β / (Conjugate[α] - α)]
```

$\left\{-\frac{\mathrm{Im}[\beta]}{2\,\mathrm{Im}[\alpha]}, \frac{\mathrm{Re}[\beta]}{2\,\mathrm{Im}[\alpha]}\right\}$

0

Factorization of motion polynomials

```
(* Factor a motion polynomial into linear factors. *)
FactorMP[(t + I) ^ 6 + eta * t]
```

$\left((i + t) + \eta \cdot \frac{1}{32}\right) \cdot \left((i + t) - \eta \cdot \frac{3}{32}\right) \cdot \left((i + t) + \eta \cdot \frac{1}{16}\right) \cdot \left((i + t) + \eta \cdot \frac{1}{16}\right) \cdot \left((i + t) - \eta \cdot \frac{3}{32}\right) \cdot \left((i + t) + \eta \cdot \frac{1}{32}\right)$

```
(* Expand again. *)
Expand[%]
```

$(-1 + 6 i t + 15 t^2 - 20 i t^3 - 15 t^4 + 6 i t^5 + t^6) + \eta \cdot t$

```
(* Now with factored coefficients *)
Factor /@ %
```

$(i + t)^6 + \eta \cdot t$

```
(* Data structure used for representing factored motion polynomials . *)
FactoredMP[eta, eta, eta]
Expand[%]
```

$(0 + \eta) \cdot (0 + \eta) \cdot (0 + \eta)$

$0 + \eta \cdot 0$

```
(* This polynomial admits several factorizations (actually 5! = 120). *)
FactorMP[poly = Product[t - k I, {k, 5}] + eta * (t + I)]
```

$$((-i+t)+\eta\cdot 0)\cdot\left((-2i+t)-\eta\cdot\frac{i}{210}\right)\cdot\left((-3i+t)+\eta\cdot\frac{3i}{280}\right)\cdot\left((-4i+t)-\eta\cdot\frac{i}{126}\right)\cdot\left((-5i+t)+\eta\cdot\frac{i}{504}\right)$$

```
(* This is an other factorization using a different order of the roots. *)
FactorMP[poly, Order → Table[k I, {k, 5, 1, -1}]]
```

$$\left((-5i+t)-\eta\cdot\frac{i}{756}\right)\cdot\left((-4i+t)+\eta\cdot\frac{i}{270}\right)\cdot((-3i+t)+\eta\cdot 0)\cdot\left((-2i+t)-\eta\cdot\frac{i}{126}\right)\cdot\left((-i+t)+\eta\cdot\frac{i}{180}\right)$$

```
(* Here not any permutation of the roots yields a factorization *)
poly = (t^2 + 1)^2 * (t + 2 I) + eta * (t + I)^2;
FactorMP[poly]
FactorMP[poly, Order → {-I, -I, I, I, -2 I}]
```

$$\left((-i+t)+\eta\cdot\left(\frac{5}{18}+C[1]\right)\right)\cdot\left((-i+t)+\eta\cdot\left(-\frac{1}{6}+C[2]\right)\right)\cdot$$
$$((i+t)-\eta\cdot C[2])\cdot((i+t)-\eta\cdot C[1])\cdot\left((2i+t)-\eta\cdot\frac{1}{9}\right)$$

Throw::nocatch : Uncaught Throw[No solution found.] returned to top level. ≫

```
Hold[Throw[No solution found.]]
```

```
(* Elliptic translational motion . *)
(* The polynomial cannot be factored, unless we multiply it by t^2+1. *)
poly = t^2 + 1 + eta * (a * t - b * I)
fact = FactorMP[poly]
```

$$(1+t^2)+\eta\cdot(-i\,b+a\,t)$$

FactorMP::R : Multiply the input with R = $1 + t^2$

$$\left((i+t)+\eta\cdot\left(\frac{a}{2}+\frac{b}{2}+C[2]\right)\right)\cdot((-i+t)-\eta\cdot C[2])\cdot\left((-i+t)+\eta\cdot\left(\frac{a}{2}-\frac{b}{2}+C[1]\right)\right)\cdot((i+t)-\eta\cdot C[1])$$

```
(* Do this factorization "by hand". *)
ansatz = FactoredMP @@ Table[t + (-1)^Ceiling[(k - 1) / 2] * I + eta * w[k], {k, 4}]
(* Derive a system of equations. *)
Union[CoefficientList[Last[Expand[ansatz] - (t^2 + 1) * poly], t]]
(* Test whether the coefficients of fact[[2]] satisfy the system . *)
Expand[% /. w[k_] :→ fact[[k, 2]]]
```

$$((i+t)+\eta\cdot w[1])\cdot((-i+t)+\eta\cdot w[2])\cdot((-i+t)+\eta\cdot w[3])\cdot((i+t)+\eta\cdot w[4])$$

$$\{i\,b-i\,w[1]-i\,w[2]+i\,w[3]+i\,w[4],\ -a+w[1]+w[2]+w[3]+w[4]\}$$

```
{0, 0}
```

```
(* All 6 factorizations of a "random " degree 3 motion polynomial . *)
roots = t /. Solve[t^3 + I == 0];
facts = FactorMP[t^3 + I + eta * (t^2 - 2), Order → #] & /@ Permutations [roots]
```

$$\left\{((-\mathrm{i}+t)+\eta\cdot 3)\cdot\left(\left(\frac{\mathrm{i}}{2}+\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{5}{4}+\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot\left(\left(\frac{\mathrm{i}}{2}-\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}-\frac{\mathrm{i}\sqrt{3}}{4}\right)\right),\right.$$

$$((-\mathrm{i}+t)+\eta\cdot 3)\cdot\left(\left(\frac{\mathrm{i}}{2}-\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{5}{4}-\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot\left(\left(\frac{\mathrm{i}}{2}+\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}+\frac{\mathrm{i}\sqrt{3}}{4}\right)\right),$$

$$\left(\left(\frac{\mathrm{i}}{2}+\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}-\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot\left((-\mathrm{i}+t)+\eta\cdot\left(\frac{5}{2}+\frac{\mathrm{i}\sqrt{3}}{2}\right)\right)\cdot\left(\left(\frac{\mathrm{i}}{2}-\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}-\frac{\mathrm{i}\sqrt{3}}{4}\right)\right),$$

$$\left(\left(\frac{\mathrm{i}}{2}+\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}-\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot\left(\left(\frac{\mathrm{i}}{2}-\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{5}{4}+\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot((-\mathrm{i}+t)+\eta\cdot 3),$$

$$\left(\left(\frac{\mathrm{i}}{2}-\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}+\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot\left((-\mathrm{i}+t)+\eta\cdot\left(\frac{5}{2}-\frac{\mathrm{i}\sqrt{3}}{2}\right)\right)\cdot\left(\left(\frac{\mathrm{i}}{2}+\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}+\frac{\mathrm{i}\sqrt{3}}{4}\right)\right),$$

$$\left.\left(\left(\frac{\mathrm{i}}{2}-\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{3}{4}+\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot\left(\left(\frac{\mathrm{i}}{2}+\frac{\sqrt{3}}{2}+t\right)+\eta\cdot\left(-\frac{5}{4}-\frac{\mathrm{i}\sqrt{3}}{4}\right)\right)\cdot((-\mathrm{i}+t)+\eta\cdot 3)\right\}$$

```
(* Show their linkages. *)
AnimateMP @@ Append[facts, Polygon → {{0, 0}, {0, 1}, {1, 1}, {1, 0}} / 2]
```

■ Animation of a motion

```
? AnimateMP
```

AnimateMP [p] animates the motion described by the motion polynomial p. If p is given in factored form , then the decomposition of the motion into revolutions is shown. AnimateMP [p1, p2, ...] shows several motions at the same time . The following options can be given:
Trace –> {x, y}: show the trace of the point {x, y} under the given motion .
Style –> "links", Style –> "wheels": different visualizations of a decomposed motion .
Polygon –> {{0, 0}, {1, 0}, {0, 1}}: specify the shape of a planar object that is displaced according to p.
Moreover, all options of the Graphics command can be used.

```
(* Animation of a simple revolution. *)
AnimateMP [t + I + 0 eta, ImageSize → {200, 200}]
```

```
(* A horizontal translational motion . *)
AnimateMP [1 + eta * (t + 1) , ImageSize → {400, 100},
  PlotRange → {{-10, 10}, {-2, 2}}, Trace → None]
```

```
(* A vertical translational motion . *)
AnimateMP [t + eta * I]
```

■ Construction of linkages

```
(* Circular translation. *)
fact = FactorMP[t^2 + 1 + eta * (t - I)]
```

$$((\mathrm{i}+t)+\eta\cdot(1+C[1]))\cdot((-\mathrm{i}+t)-\eta\cdot C[1])$$

```
(* Show different decompositions of this motion depending on the choice of C[1]. *)
AnimateMP @@ Table[fact /. C[1] → k, {k, -1 / 2, 1, 1 / 2}]
```

```
(* Define a linkage by constructing its link graph. *)
linkage= {{1, 2, fact[[2]] /. C[1] → -1 / 2}, {1, 3, fact[[2]] /. C[1] → 0},
    {2, 4, fact[[1]] /. C[1] → -1 / 2}, {3, 4, fact[[1]] /. C[1] → 0}}
```

$$\left\{\left\{1,2,(-\mathrm{i}+t)+\eta\cdot\frac{1}{2}\right\},\{1,3,(-\mathrm{i}+t)+\eta\cdot 0\},\left\{2,4,(\mathrm{i}+t)+\eta\cdot\frac{1}{2}\right\},\{3,4,(\mathrm{i}+t)+\eta\}\right\}$$

```
? ShowLinkage
```

ShowLinkage[graph, base] creates an animation of a linkage. The link graph must be given in the form {{i, j, mij }, ...} where i and j are the labels of two links and mij is a linear bounded motion polynomial describing the relative position between these two links. The second argument, base, is the label of the base link (the one that doesn't move). The following options can be given:
ColorOutput –> {1 –> Yellow, 2 –> Green, ...}: to draw link 1 in yellow etc.
Thickness –> 0.01: thickness of links. Thickness –> {0.01, 0.001}: thickness of links and lines.
Trace –> {i, {x, y}}: trace the point {x, y} under the motion of link i.
Direction –> {x, y}: draw the curve with a pen whose shape is a line from {0, 0} to {x, y}.
Range –> {a, b}: range for the time ArcTan[t], i.e., [a, b] ⊆ [–π/2, π/2].
Links –> {i, j, ...}: specify an order in which the links should be drawn.
Style –> "CompleteGraph", Style –> "Star": how to draw links that are connected with more than 2 joints.
Return –> "animation": produces an animation of the linkage.
Return –> {"picture", x}: produces an image showing the linkage at time x = ArcTan[t].
Return –> {"movie", n}: creates a sequence of n image files that can be used to make a movie.
Return –> {"movie", {t1, t2, ..., tn}}: as before, using time ti for frame i.
Return –> "graph": draws the link graph.
Return –> "collisions": returns a list of the form {{{i, k, j}, s, t}, ...} which means that the joint connecting i and j collides with link k at time t at position 0 <= s <= 1.
Return works similarly with "animation3D", "picture3D", and "movie3D".
Moreover, all options of the Graphics / Graphics3D commands can be used.

```
ShowLinkage[linkage, 1, Trace→{4, {0, 0}}, Thickness→{0.007, 0.003}, ImageSize →200]

(* A degree 6 motion polynomial that allows only a single factorization *)
mp1 = FactoredMP@@Table[t + I + eta * k * I, {k, 6}]
AnimateMP [mp1 , Polygon→{{0, 0}}]
```

$((i+t)+\eta\cdot i)\cdot((i+t)+\eta\cdot(2i))\cdot((i+t)+\eta\cdot(3i))\cdot((i+t)+\eta\cdot(4i))\cdot((i+t)+\eta\cdot(5i))\cdot((i+t)+\eta\cdot(6i))$

```
(* A different polynomial that produces basically the same curve. *)
mp2 = FactoredMP@@Table[t + I + eta * (1 + (-1) ^k) * I, {k, 0, 5}]
AnimateMP [mp2 , Polygon→{{0, 0}}]
```

$((i+t)+\eta\cdot(2i))\cdot((i+t)+\eta\cdot 0)\cdot((i+t)+\eta\cdot(2i))\cdot((i+t)+\eta\cdot 0)\cdot((i+t)+\eta\cdot(2i))\cdot((i+t)+\eta\cdot 0)$

```
? ConstructLinkage
```

ConstructLinkage[mp, l1] constructs a linkage of mobility one that realizes the motion given by mp. The motion polynomial mp must be given in factored form, and l1 is a linear bounded motion polynomial. If l1 is omitted, a suitable such polynomial is taken instead. The output is given in the form as needed for ShowLinkage.

```
linkage= ConstructLinkage[mp2 , t + 2 I + 0 * eta]
```

$\{\{1, 2, (i+t)+\eta\cdot 0\}, \{2, 3, (i+t)+\eta\cdot(2i)\}, \{3, 4, (i+t)+\eta\cdot 0\}, \{4, 5, (i+t)+\eta\cdot(2i)\}, \{5, 6, (i+t)+\eta\cdot 0\}, \{6, 7, (i+t)+\eta\cdot(2i)\}, \{8, 9, (i+t)+\eta\cdot\frac{1456i}{729}\}, \{9, 10, (i+t)-\eta\cdot\frac{2i}{243}\}, \{10, 11, (i+t)+\eta\cdot\frac{160i}{81}\}, \{11, 12, (i+t)-\eta\cdot\frac{2i}{27}\}, \{12, 13, (i+t)+\eta\cdot\frac{16i}{9}\}, \{13, 14, (i+t)-\eta\cdot\frac{2i}{3}\}, \{1, 8, (2i+t)+\eta\cdot\frac{728i}{729}\}, \{2, 9, (2i+t)+\eta\cdot\frac{728i}{243}\}, \{3, 10, (2i+t)+\eta\cdot\frac{80i}{81}\}, \{4, 11, (2i+t)+\eta\cdot\frac{80i}{27}\}, \{5, 12, (2i+t)+\eta\cdot\frac{8i}{9}\}, \{6, 13, (2i+t)+\eta\cdot\frac{8i}{3}\}, \{7, 14, (2i+t)+\eta\cdot 0\}\}$

```
ShowLinkage[linkage, PlotRange→{{-2, 2}, {-1, 2.5}}, Trace→{7, {0.1, 0}}]
```

Example 4.4 from the paper

```
P1 = t + I - 3 eta;
P3 = t + 2 I + eta * (18 + 3 I);
{P2, P4} = Flip[P1, P3];
TableForm [{P1, P3, P4, P2}]
```

```
(i + t) - η · 3
(2 i + t) + η · (18 + 3 i)
(i + t) + η · (13 + 2 i)
(2 i + t) + η · (2 + i)
```

```
FixPoint /@ {P1, P3, P4, P2}
```

$$\left\{\left\{0, -\frac{3}{2}\right\}, \left\{-\frac{3}{4}, \frac{9}{2}\right\}, \left\{-1, \frac{13}{2}\right\}, \left\{-\frac{1}{4}, \frac{1}{2}\right\}\right\}$$

```
(* Check cycle condition (with quick-and-dirty way of inverting P1 and P3). *)
P2 ** P4 ** (2 t - P3) ** (2 t - P1)
```

$(4 + 5 t^2 + t^4) + \eta \cdot 0$

```
linkage = {{1, 2, P2}, {3, 4, P3}, {1, 3, P1}, {2, 4, P4}}
```

```
{{1, 2, (2 i + t) + η · (2 + i)}, {3, 4, (2 i + t) + η · (18 + 3 i)},
  {1, 3, (i + t) - η · 3}, {2, 4, (i + t) + η · (13 + 2 i)}}
```

```
ConstructLinkage[FactoredMP[P2], P4] === linkage
```

```
True
```

```
ShowLinkage[linkage, Axes → True,
  ColorOutput → {1 → Yellow, 2 → Red, 3 → Green, 4 → Blue}, Thickness → 0.1, Links → {2, 1, 4, 3}]

ShowLinkage[linkage, ColorOutput → {1 → Yellow, 2 → Red, 3 → Green, 4 → Blue},
  Thickness → 0.1, Links → {2, 1, 4, 3}, Return → "animation3D"]

opts = Sequence @@ {PlotRange → {{-1, 6}, {-2, 4}}, Axes → True,
        ImageSize → 1000, LabelStyle → 40, Ticks → {Automatic , {-2, 0, 1, 2, 3, 4}}};
pic[l3_] := ShowLinkage[ReplacePart[linkage, {2, 1} → l3], Return → {"picture", -1.10715},
      ColorOutput → {1 → Black, 4 → White}, Thickness → 0.04, opts];
TableForm [{{Show[pic[5], ImageSize → 200, LabelStyle → 10],
      Show[pic[3], ImageSize → 200, LabelStyle → 10]}}]
(*
Export["../images /weak.jpg",pic[5],"CompressionLevel "→0.1];
Export["../images /strong.jpg",pic[3],"CompressionLevel "→0.1];
*)
```

Example: a simple quadratic motion polynomial

```
(* A quadratic motion . *)
poly = (t + I) * (t - 1 + I) + eta * (t + 2 I)
AnimateMP [poly]
```

$((-1 - i) - (1 - 2 i) t + t^2) + \eta \cdot (2 i + t)$

```
(* Now factor this quadratic polynomial  into linear factors. *)
fact = FactorMP[poly]
```

$$\left((i + t) + \eta \cdot \left(\frac{6}{5} - \frac{3 i}{5}\right)\right) \cdot \left(((-1 + i) + t) + \eta \cdot \left(-\frac{1}{5} + \frac{3 i}{5}\right)\right)$$

```
(* Test whether the factorization is correct. *)
Expand[fact] - poly
```

$0 + \eta \cdot 0$

```
(* Animation of the factored motion . *)
AnimateMP [fact]
```

```
(* Different style of animation. *)
AnimateMP [fact, Style→"wheels"]
```

```
(* Compute  the second factorization *)
facts = {fact, FactorMP[poly, Order→{1 - I, -I}]}
```

$$\left\{\left((i + t) + \eta \cdot \left(\frac{6}{5} - \frac{3\,i}{5}\right)\right) \cdot \left(((-1 + i) + t) + \eta \cdot \left(-\frac{1}{5} + \frac{3\,i}{5}\right)\right),\ \left(((-1 + i) + t) + \eta \cdot \left(\frac{7}{5} + \frac{i}{5}\right)\right) \cdot \left((i + t) + \eta \cdot \left(-\frac{2}{5} - \frac{i}{5}\right)\right)\right\}$$

```
(* Show the corresponding linkage. *)
{{P1, P2}, {P3, P4}} = List@@@facts;
link= {{2, 1, P1}, {3, 2, P2}, {3, 4, P4}, {4, 1, P3}};
ShowLinkage[link, 3, Trace→{1, {0, 0}}, Thickness→{0.015, 0.003}, Links→{4, 1, 3, 2}]
```

```
(* Create movie  ToyExample .mp4  *)
ShowLinkage[link, 3, Trace→{1, {0, 0}}, Thickness→{0.015, 0.003},
  Links→{4, 1, 3, 2}, ImageSize →1000, Return→{"movie ", 150}]
```

```
In the directory /win/d/Forschung/Publikationen/15_PlanarLinkages/software
    execute the following command  :
ffmpeg  -i temp %05d.jpg -r 25 -qscale 4 movie .mp4
```

Example: elliptic translation

```
(* We consider the ellipse given by the equation (x+1)^2 + 4*y^2 == 1. *)
ContourPlot[(x+1)^2+4y^2==1, {x, -2, 0}, {y, -1/2, 1/2}, AspectRatio→1/2]
```



```
(* We use the following parametrization. *)
ParametricPlot [{-2, t} / (t^2+1), {t, -100, 100}, PlotRange→{{-2, 0}, {-1/2, 1/2}}]
```




```
(* This translates into the following motion polynomial , *)
(* realizing a translational motion along the ellipse. *)
poly=t^2+1+eta*(-2+I*t);
AnimateMP [poly]
```

```
(* To factor this motion polynomial , it is multiplied by t^2+1. *)
(* We obtain a 2-parameter family of factorizations *)
fact1=FactorMP[poly]
```

FactorMP::R : Multiply the input with R = $1 + t^2$

$$\left((i+t)+\eta\cdot\left(-\frac{i}{2}+C[2]\right)\right)\cdot((-i+t)-\eta\cdot C[2])\cdot\left((-i+t)+\eta\cdot\left(\frac{3\,i}{2}+C[1]\right)\right)\cdot((i+t)-\eta\cdot C[1])$$

```
(* This is a factorization with different order of the roots. *)
FactorMP[poly, Order→{I, -I, -I, I}]
```

FactorMP::R : Multiply the input with R = $1 + t^2$

$$\left((-i+t)+\eta\cdot\left(\frac{3\,i}{2}+C[2]\right)\right)\cdot((i+t)-\eta\cdot C[2])\cdot\left((i+t)+\eta\cdot\left(-\frac{i}{2}+C[1]\right)\right)\cdot((-i+t)-\eta\cdot C[1])$$

```
(* Using the first factorization, we can construct a linkage which *)
(* strongly realizes the translational motion along the ellipse. *)
ShowLinkage[ConstructLinkage[fact1/.{C[1]→0, C[2]→-I/2}, t+9/5I+eta*0],
  Trace→{5, {0, 0}}, Direction→{0.03, 0}]
```

```
(* Multiplying with t-I from the left also enables us to find a factorization *)
(* By doing so we change the motion (it is not any more a translational one). *)
(* However, the orbit of the origin under this new motion is still the ellipse. *)
fact2=FactorMP[(t-I)**(t^2+1+eta*(-2+I*t))]
```

$$\left((-i+t)-\eta\cdot\frac{i}{2}\right)\cdot\left((-i+t)+\eta\cdot\left(\frac{3\,i}{2}+C[1]\right)\right)\cdot((i+t)-\eta\cdot C[1])$$

```
(* Choose some value for the parameter of this one-
  dimensional family of factorizations *)
fact2=fact2/.C[1]→-I
```

$$\left((-i+t)-\eta\cdot\frac{i}{2}\right)\cdot\left((-i+t)+\eta\cdot\frac{i}{2}\right)\cdot((i+t)+\eta\cdot i)$$

```
(* Use this factorization to construct a linkage. *)
(* The second argument is a linear motion
  polynomial that can be chosen almost randomly . *)
(* With some experiments , we found this one which gives a nice-looking linkage. *)
linkEll= ConstructLinkage[fact2, t + 9 / 5 * I + eta * 18 / 35 * I]
```

$\{\{1, 2, (i + t) + \eta \cdot i\}, \{2, 3, (-i + t) + \eta \cdot \frac{i}{2}\}, \{3, 4, (-i + t) - \eta \cdot \frac{i}{2}\}, \{5, 6, (i + t) + \eta \cdot \frac{11 i}{56}\},$
$\{6, 7, (-i + t) - \eta \cdot \frac{5 i}{8}\}, \{7, 8, (-i + t) + \eta \cdot \frac{13 i}{28}\}, \{1, 5, (\frac{9 i}{5} + t) + \eta \cdot \frac{207 i}{140}\},$
$\{2, 6, (\frac{9 i}{5} + t) + \eta \cdot \frac{27 i}{40}\}, \{3, 7, (\frac{9 i}{5} + t) - \eta \cdot \frac{9 i}{20}\}, \{4, 8, (\frac{9 i}{5} + t) + \eta \cdot \frac{18 i}{35}\}\}$

```
(* Rename the labels such that they agree with those in the paper. *)
linkEll= Map[If[IntegerQ[#], If[# ≤ 4, 5 - #, 13 - #], #] &, linkEll, {2}];

(* Define some coloring of the links. *)
colEll = {4 → White, 3 → Yellow, 2 → Orange, 1 → Red, 8 → Green, 7 → Cyan, 6 → Blue, 5 → Magenta};

(* Now the motion produced by the end effector is not any more translational. *)
ShowLinkage[linkEll, 4, Trace → {1, {0, 0}}, ColorOutput → colEll, Direction → {0.03, 0}]

(* Define an order of the links. This produces 5 collisions at t=∞. *)
(* This is the ordering produced by our code in ConstructLinkage. *)
(* Also the pen collides only at t=∞. *)
ordEll = {8, 4, 5, 1, 6, 2, 7, 3};
ShowLinkage[linkEll, 4, Trace → {1, {0, 0}}, Links → ordEll, Return → "collisions"]
ShowLinkage[linkEll, 4, Trace → {1, {0, 0}}, Links → ordEll, ColorOutput → colEll]
```

$\{\{\{8, 5, 7\}, \frac{5}{42}, \infty\}, \{\{8, 1, 7\}, \frac{11}{28}, \infty\}, \{\{8, 1, 7\}, \frac{11}{16}, \infty\}, \{\{8, 6, 7\}, \frac{37}{61}, \infty\}, \{\{8, 6, 7\}, \frac{25}{49}, \infty\},$
$\{\{8, 2, 7\}, \frac{17}{56}, \infty\}, \{\{8, 2, 7\}, \frac{25}{42}, \infty\}, \{\{1, 6, 2\}, \frac{54}{61}, \infty\}, \{\{1, 6, 2\}, \frac{6}{7}, \infty\}\}$

```
(* Here is a different ordering, figured out by hand. *)
(* It produces only 2 collisions, both at t=∞. *)
(* Also the pen collides only at t=∞. *)
ordEll = {5, 1, 6, 2, 7, 8, 4, 3};
ShowLinkage[linkEll, 4, Trace → {1, {0, 0}}, Links → ordEll, Return → "collisions"]
ShowLinkage[linkEll, 4, Trace → {1, {0, 0}}, Links → ordEll,
  ColorOutput → colEll, ImageSize → 600, Return → "animation3D"]
```

$\{\{\{1, 6, 2\}, \frac{54}{61}, \infty\}, \{\{1, 6, 2\}, \frac{6}{7}, \infty\}, \{\{7, 8, 3\}, \frac{2}{7}, \infty\}\}$

```
ellgraph[s_] := ShowLinkage[linkEll, 4, ColorOutput → colEll,
      Thickness → {1, 0.004}, Return → "graph", ImageSize → s];
ellgraph[400]
(*Export["../www/EllipseLG.jpg",ellgraph[600],"CompressionLevel "→0.01];*)
```



```
(* In order to build the linkage, compute integral lengths of the links. *)
(* For each link the positions of the joints connected to it are given, *)
(* namely their x-coordinates in the initialconfiguration of the linkage. *)
pos = First/@ (112 * (FixPoint[Last[#]] & /@ linkEll));
Table[pos[[First/@ Position[Most /@ linkEll, i]]], {i, 8}]
```

```
{{-28, -16}, {28, -28, 14}, {-56, 28, -21},
  {-56, -46}, {26, -16}, {-35, 26, 14}, {-11, -35, -21}, {-11, -46}}
```

```
(* Create movie Ellipse.mp4 . *)
ShowLinkage[linkEll, 4, Trace → {1, {0, 0}}, Links → ordEll, ColorOutput → colEll,
  Range → {-Pi/2 + 10^(-6), Pi/2 + 1/2}, ImageSize → 2000, Return → {"movie", 300}]
```

```
In the directory /win/d/Forschung/Projekte/motion-poly/software
    execute the following command :
ffmpeg -i temp %05d.jpg -r 25 -qscale 4 movie.mp4
```

```
(* Create movie Ellipse3D.mp4 . *) ShowLinkage[linkEll, 4, Trace → {1, {0, 0}},
  Links → ordEll, ColorOutput → colEll, Boxed → False, Thickness → {0.007, 0.004},
  ImageSize → 2000, PlotRange → {{-0.65, 0.3}, {-0.43, 0.5}}, Range → {1.3, 1.8},
  ViewPoint → {-0.1, -3, 1.5}, ViewVertical → {0, 0, 1}, Return → {"movie3D", 200}]
```

```
In the directory /win/d/Forschung/Projekte/motion-poly/software
    execute the following command :
ffmpeg -i temp %05d.jpg -r 25 -qscale 4 movie.mp4
```

```
(* Create the picture shown in the paper. *)
grpr = MapThread[First[ShowLinkage[linkEll, 4, Trace → {1, {0, 0}},
            ColorOutput → {_ → #2}, Links → ordEll, Return → {"picture", #1}]] &,
      {ArcTan[{2, 1/2, 0, -1}], {White, LightGray, Gray, Black}}];
cnt = 1;
grpr = Flatten[grpr /. {Red → Black, RGBColor[0.99, 0, 0] → White} /.
      Line[a_List /; Length[a] > 10] :> If[cnt == 1, cnt++;
          {Thickness[0.006], Line[a], Thickness[0.002]}, {}]];
Clear[cnt];
Graphics[grpr, PlotRange → {{-2.05, 0.05}, {-0.9, 0.75}}, ImageSize → 600]
(*
Export["../images/elliptic.jpg",Graphics[grpr,
      PlotRange→{{-2.05,0.05},{-0.9,0.75}},ImageSize→2000],"CompressionLevel"→0.1];
*)
```

```
(* Construct a collision-free linkage using some links of special shape. *)
cfEll = Join[linkEll, Table[{i, -i, FixPoint[Cases[linkEll, {i-4, i, a_} → a][[1]]]}, {i, 6, 7}]];
cfEll = cfEll /. {a : (6 | 7), b_, c_} /; a - b == 1 → {-a, b, c};
cfEll = cfEll /. {a : (2 | 3 | 4), b_, c_} /; b - a == 4 → {-a, b, c};
cfEll = Join[cfEll, {{2, -2, {-3/5, 0}}, {3, -3, {-7/10, 0}}, {4, -4, {7/20, 0}}}];
cfCol = Join[colEll, -#1 → #2 & @@@ colEll] /. White → GrayLevel[0.8];
```

```
(* Display the collision-free linkage. *)
ShowLinkage[cfEll, 4, Trace → {1, {0, 0}}, Links → {-4, 8, 7, -3, -7, 6, -2, -6, 5, 1, 2, 3, 4},
  ColorOutput → cfCol, Boxed → False, Thickness → {0.011, 0.004}, Return → "animation3D",
  ViewPoint → {-0.1, -3, 1.5}, ViewVertical → {0, 0, 1}, ImageSize → 600,
  Lighting → {{"Directional", GrayLevel[0.85], 10 {2, -3, 3}}, {"Directional",
       GrayLevel[0.3], 10 {-2, -3, 3}}, {"Directional", GrayLevel[0.2], 10 {0, -3, -3}}}]
```

Example: John Hancock's "J"

```
(* Our parametrization {x/d, y/d} of Hancock's J. *)
With[{
   x = -451 - 8666 t - 64488 t^2 - 237449 t^3 - 436132 t^4 - 321880 t^5,
   y = -651 - 11325 t - 78158 t^2 - 270569 t^3 - 472949 t^4 - 336018 t^5,
   d = 170 * (9 + 162 t + 1219 t^2 + 4908 t^3 + 11187 t^4 + 13770 t^5 + 7225 t^6)},
  poly = (d + eta ** (x + I * y)) / Coefficient[d, t^6]]
```

$$\left(\frac{9}{7225}+\frac{162\,t}{7225}+\frac{1219\,t^2}{7225}+\frac{4908\,t^3}{7225}+\frac{11187\,t^4}{7225}+\frac{162\,t^5}{85}+t^6\right)+$$
$$\eta\cdot\left(\left(-\frac{451}{1228250}-\frac{651\,i}{1228250}\right)-\left(\frac{4333}{614125}+\frac{453\,i}{49130}\right)t-\left(\frac{32244}{614125}+\frac{39079\,i}{614125}\right)t^2-\right.$$
$$\left.\left(\frac{237449}{1228250}+\frac{270569\,i}{1228250}\right)t^3-\left(\frac{218066}{614125}+\frac{472949\,i}{1228250}\right)t^4-\left(\frac{32188}{122825}+\frac{168009\,i}{614125}\right)t^5\right)$$

```
fact = FactorMP[poly]
```

FactorMP::R : Multiply the input with R = $\frac{9}{7225}+\frac{162\,t}{7225}+\frac{1219\,t^2}{7225}+\frac{4908\,t^3}{7225}+\frac{11187\,t^4}{7225}+\frac{162\,t^5}{85}+t^6$

$$\left(\left(\left(\frac{2}{5}+\frac{i}{5}\right)+t\right)+\eta\cdot\left(\left(-\frac{7309}{36125}-\frac{17059\,i}{1156000}\right)+C[6]\right)\right)\cdot\left(\left(\left(\frac{2}{5}-\frac{i}{5}\right)+t\right)-\eta\cdot C[6]\right)\cdot$$
$$\left(\left(\left(\frac{2}{5}-\frac{i}{5}\right)+t\right)+\eta\cdot\left(\left(-\frac{13263}{289000}-\frac{228789\,i}{1156000}\right)+C[5]\right)\right)\cdot\left(\left(\left(\frac{2}{5}+\frac{i}{5}\right)+t\right)-\eta\cdot C[5]\right)\cdot$$
$$\left(\left(\left(\frac{27}{85}+\frac{6\,i}{85}\right)+t\right)+\eta\cdot\left(\left(\frac{75057}{4913000}+\frac{743611\,i}{29478000}\right)+C[4]\right)\right)\cdot\left(\left(\left(\frac{27}{85}-\frac{6\,i}{85}\right)+t\right)-\eta\cdot C[4]\right)\cdot$$
$$\left(\left(\left(\frac{27}{85}-\frac{6\,i}{85}\right)+t\right)+\eta\cdot\left(\left(\frac{64739}{2456500}-\frac{980959\,i}{29478000}\right)+C[3]\right)\right)\cdot\left(\left(\left(\frac{27}{85}+\frac{6\,i}{85}\right)+t\right)-\eta\cdot C[3]\right)\cdot$$
$$\left(\left(\left(\frac{4}{17}+\frac{i}{17}\right)+t\right)+\eta\cdot\left(\left(-\frac{124863}{1965200}+\frac{46513\,i}{3930400}\right)+C[2]\right)\right)\cdot\left(\left(\left(\frac{4}{17}-\frac{i}{17}\right)+t\right)-\eta\cdot C[2]\right)\cdot$$
$$\left(\left(\left(\frac{4}{17}-\frac{i}{17}\right)+t\right)+\eta\cdot\left(\left(\frac{15839}{1965200}-\frac{254241\,i}{3930400}\right)+C[1]\right)\right)\cdot\left(\left(\left(\frac{4}{17}+\frac{i}{17}\right)+t\right)-\eta\cdot C[1]\right)$$

```
(* Choose values for the free parameters . *)
fact = fact /. Table[C[i] → C[i] - fact[[13 - 2 * i, 2]], {i, 6}]
```

$$\left(\left(\left(\frac{2}{5}+\frac{i}{5}\right)+t\right)+\eta\cdot 0\right)\cdot\left(\left(\left(\frac{2}{5}-\frac{i}{5}\right)+t\right)+\eta\cdot\left(-\frac{7309}{36125}-\frac{17059\,i}{1156000}\right)\right)\cdot$$
$$\left(\left(\left(\frac{2}{5}-\frac{i}{5}\right)+t\right)+\eta\cdot 0\right)\cdot\left(\left(\left(\frac{2}{5}+\frac{i}{5}\right)+t\right)+\eta\cdot\left(-\frac{13263}{289000}-\frac{228789\,i}{1156000}\right)\right)\cdot$$
$$\left(\left(\left(\frac{27}{85}+\frac{6\,i}{85}\right)+t\right)+\eta\cdot 0\right)\cdot\left(\left(\left(\frac{27}{85}-\frac{6\,i}{85}\right)+t\right)+\eta\cdot\left(\frac{75057}{4913000}+\frac{743611\,i}{29478000}\right)\right)\cdot$$
$$\left(\left(\left(\frac{27}{85}-\frac{6\,i}{85}\right)+t\right)+\eta\cdot 0\right)\cdot\left(\left(\left(\frac{27}{85}+\frac{6\,i}{85}\right)+t\right)+\eta\cdot\left(\frac{64739}{2456500}-\frac{980959\,i}{29478000}\right)\right)\cdot$$
$$\left(\left(\left(\frac{4}{17}+\frac{i}{17}\right)+t\right)+\eta\cdot 0\right)\cdot\left(\left(\left(\frac{4}{17}-\frac{i}{17}\right)+t\right)+\eta\cdot\left(-\frac{124863}{1965200}+\frac{46513\,i}{3930400}\right)\right)\cdot$$
$$\left(\left(\left(\frac{4}{17}-\frac{i}{17}\right)+t\right)+\eta\cdot 0\right)\cdot\left(\left(\left(\frac{4}{17}+\frac{i}{17}\right)+t\right)+\eta\cdot\left(\frac{15839}{1965200}-\frac{254241\,i}{3930400}\right)\right)$$

```
(* Construct the linkage and show the link graph. *)
linkJ= ConstructLinkage[fact, t+1+I+eta];
ShowLinkage[linkJ, Return→"graph", Thickness→{0, 0.001}, ImageSize →650]
```




```
(* Some  options for drawing this linkage *)
optsJ = Sequence[Trace→{13, {-7/10, 7/10}}, Direction→{0.025, 0.03},
      Range→{-0.55, 1.7}, ImageSize →500, Thickness→{0.009, 0.001},
      ColorOutput→Flatten[Table[Thread[{i, i+13}→GrayLevel[1/2+i/26]], {i, 13}]],
      Links→{2, 1, 14, 15, 3, 16, 4, 5, 17, 18, 6, 19,
          7, 20, 8, 21, 22, 9, 23, 10, 11, 24, 25, 12, 26, 13}];

(* 2D animation *)
ShowLinkage[linkJ, Axes→True, optsJ]

(* 3D animation using a different style for the links *)
ShowLinkage[linkJ, Axes→True, ColorOutput→{}, Return→"animation3D", Style→"Star", optsJ]

(* 3D animation with a different ordering of the links *)
ShowLinkage[linkJ, Axes→True, ColorOutput→{},
  Links→{14, 1, 17, 4, 18, 5, 21, 8, 22, 9, 25, 12, 26, 13, 24, 11, 23, 10, 20, 7, 19, 6, 16, 3, 15, 2},
  Return→"animation3D", optsJ]

(* Produce the picture shown in the paper. *)
(* Use "/200" for the steps in TraceMP. *)
Get["PlanarLinkages.m "];
picJ= ShowLinkage[linkJ, Return→{"picture", -0.325},
      ImageSize →4000, PlotRange→{{-1.1, 2.4}, {-0.1, 2.2}}, optsJ];
picJ= picJ/.{Red→Black, RGBColor[0.99, 0, 0]→White};
Show[picJ, ImageSize →500]
(*Export["../images /JohnHancockLink.jpg",
    ImageResize [picJ,4000],"CompressionLevel "→0.01];*)

(* The speed of the pen is not very uniform . *)
expr = Norm [D[ActR2[poly, {0, 0}], t]];
Plot[expr*If[t>0, Exp[t], 1], {t, -0.55, 1.7}, PlotRange→{0, 50}]
```



```
(* Use non-constant time steps. *)
TimeDiscretization = {}; t0 = -0.55; speed = {};
While[t0 < 1.7, With[{d = 0.003 * (2 Erf[20 (t0 + 0.1)] + 3) * (Erf[-6 (t0 - 3 / 2)] / 2 + 1 / 2) ^ 0},
      AppendTo[TimeDiscretization, t0 += d];
      AppendTo[speed, {t0, d * (expr /. t → t0)}]]];
Print["Length: ", Length[TimeDiscretization]];
ListPlot[speed, Joined → True, PlotRange → {0, 0.2}]
```

```
Length: 263
```




```
(* Generate pictures for movie HancockJ.mp4 . *)
ShowLinkage[linkJ, Return → {"movie", TimeDiscretization},
  ImageSize → 1400, PlotRange → {{-1.1, 3}, {-0.8, 2.4}}, optsJ]
```

```
In the directory /win/d/Forschung/Projekte/motion-poly/software
    execute the following command :
ffmpeg -i temp%05d.jpg -r 25 -qscale 4 movie.mp4
```

```
(* If we stack the links according to their labels, we get lots of collisions *)
Timing[Length[ShowLinkage[linkJ, Links → Range[26], Return → "collisions"]]]
```

```
{72.312000, 240}
```

```
coll = ShowLinkage[linkJ, Return → "collisions", optsJ];
Print["All collisions: ", Length[coll]];
coll = Select[coll, ArcTan[-0.55] < Last[#] < ArcTan[1.7] &];
Print["Collisions in range: ", Length[coll]];
Print["  counting triangles: ", Length[Union[First /@ coll]]];
{#1, N[#2], N[#3]} & @@@ coll
```

```
All collisions: 28

Collisions in range: 11

  counting triangles: 7

{{{5, 17, 18}, 0.769094, 0.0630539},
  {{5, 18, 6}, 0.908195, -0.315322}, {{18, 6, 19}, 0.908195, -0.315322},
  {{18, 6, 19}, 0.450507, 0.0914514}, {{8, 21, 9}, 0.87135, 0.114051},
  {{8, 21, 9}, 0.908195, -0.315322}, {{8, 22, 9}, 0.655638, 0.0748382},
  {{8, 22, 9}, 0.652891, 0.842441}, {{23, 10, 24}, 0.745356, -0.381966},
  {{11, 25, 12}, 0.888257, -0.203819}, {{11, 25, 12}, 0.745356, -0.381966}}
```

```
(* Observe the 2nd and 3rd collision in the animation. *)
ShowLinkage[linkJ, Range → ArcTan[coll[[2, 3]] + {-0.01, 0.01}], ImageSize → 500,
  Thickness → 0.001, PlotRange → {{1.0, 1.3}, {0.9, 1.2}, {0.0, 0.15}},
  ColorOutput → {5 → Green, 18 → Blue, 6 → Yellow, 19 → Red}, Return → "animation3D", optsJ]
```

```
(* This ordering of the links has only 6 collisions, but the pen is not on top. *)
Timing[
  Union[First/@Select[ShowLinkage[linkJ, Links→{2, 1, 14, 15, 3, 16, 4, 5, 17, 18, 6, 19, 7, 20,
             8, 21, 22, 9, 23, 10, 11, 24, 12, 13, 25, 26},
        Return→"collisions", optsJ], ArcTan[-0.55] < Last[#] < ArcTan[1.7] &]]]
```

```
{10.139000, {{5, 17, 18}, {5, 18, 6}, {8, 21, 9}, {8, 22, 9}, {18, 6, 19}, {23, 10, 24}}}
```